Nicholas L. Georgakopoulos

A Model of Reactions to Vague Limits

This text is the mathematical appendix or explanation to my article, *The Vagueness of Limits and the Desired Distribution of Conducts*, which appears in volume 32 of the CONNECTICUT LAW REVIEW, 1999.

Individuals have utility functions **u(·)** which peak at their preferences **p**: $u(c, p)=-s(p-c)^2$. The factor **s** is simply used to adjust the intensity of the preferences. It is the same for all individuals regardless of preference.

In[1]:=

```
u[c_,p_]:=-s(p-c)^2
```

A violation (actually, one that is certainly a violation as opposed to one within the vagueness range) produces a penalty **p(·)** which is dependent on the size of the violation. In the context of a vague limit that will require as inputs the realization of the limit **t** and the conduct **c** using some multiplier **m** for its intensity: $p(c,t)=m(c-t)$.

In[2]:=

```
p[c_,t_]:=m(c-t)
```

The reaction to a precise penalty is a simple optimization. Individuals who will choose to violate will choose a conduct $c=p-m/2s$. The following command solves the optimization problem by taking the derivative of utility minus penalty with respect to conduct, equating it to zero, and solving for conduct. The resulting equation is assigned to function *cdis[ ]*:

In[52]:=

```
cdis[n_,x_,p_,m_,s_]=c/.Solve[D[u[c,p]-
p[c,t],c]==0,c][[1]]//Simplify
```

Out[52]=

```
      m
p - ---
    2 s
```

To make the transition to vague limits, we must think about what such individuals, who will disregard the limit, will do if the limit is vague. If they are choosing to violate with certainty a vague limit, they will use the expectation of the limit as the penalty. Thus, the above reaction will remain the same.

The vague limit, however, does not produce only the group who abide by the limit and a group who violate by disregarding the limit. A third group exists, those who chose conducts inside the range where the vague limit occurs. In other words, these choose conducts which are sometimes violations and sometimes not, depending on the vague limit's realization.

While the normal distribution would be most appropriate to model the type of uncertainty involved in vague limits, it makes the analysis intractable. We will approximate the normal using the triangular distribution. The probability density function of the triangular distribution looks like a tent or an isosceles triangle. Its range is from a minimum n to a maximum x. It has its mean and mode at the midpoint, (x+n)/2, and is symmetrical. The probability density function below the mean is given by $g(z)= 4(z-n)/(x-n)^2$ and above the mean by $g(z|z>(x+n)/2)= 4(x-z)/(x-n)^2$.

For the individual who chooses a conduct in the lower half of the distribution of the limit, the penalty will take a probabilistic form. It will be the integration of all possible penalties that may be suffered if a realization of the limit occurs from the minimum **n** to the conduct **c**:

$$Ep(c) = \int_n^c g(z)m(c-z)dz$$

The optimization problem, however, is similar and solvable. The following command derives the derivative of utility minus penalty with respect to conduct, equates that to zero, solves for conduct, selects the second of the two roots (solutions), simplifies it, and assigns it to the function *clow[ ]*:

In[23]:=

```
clow[n_,x_,p_,m_,s_]=c/.Solve[ D[u[c,p]-Integrate[
4(z-n)/(x-n)^2 m(c-z),{z,n,c}],c]==0,c][[2]]//
PowerExpand//Simplify
```

Out[23]=

```
            2                                    2         2
(2 m n - n  s + Sqrt[s (-4 m (n - p) + s (n - x) ) (n - x) ] +

                  2
   2 n s x - s x ) / (2 m)
```

For example, if the limit is 65±5 (i.e. the vagueness ranges from 60 to 70), someone who would have chosen 65 will chose 63.66:

In[24]:=

```
clow[60,70,65,1,.1]
```

Out[24]=

```
63.6603
```

More complex is the determination of the reaction for individuals who choose conducts in the upper half of the vagueness range. Their expected penalty for realization of the limit in the lower half of the distribution can be determined probabilistically as their response to the precise limit equal to the truncated mean given a realization in the lower half of the limit.

Since the truncation is at the mid-point of the triangular distribution we can multiply by 2 instead of the cumulative density function the integration of the limit up to the mean:

$$E(t|t<\tfrac{n+x}{2}) = 2\int_{n}^{\frac{n+x}{2}} g(z)zdz$$

In[13]:=

```
2 Integrate[4(z-n)/(x-n)^2 z,{z,n,(x+n)/2}]//
PowerExpand//Simplify
```

Out[13]=

```
2 n + x
-------
   3
```

Since the truncated limit is (2n+x)/3 the expected penalty is half the time based on that and the rest of the time the integral from there on, namely:

$$E(p|\tfrac{x+n}{2}<c<x) = .5m(c-\tfrac{2n+x}{3}) + \int_{\frac{x+n}{2}}^{c} g(z)m(c-z)dz$$

The optimization problem is still solvable. The following command derives the derivative of utility minus penalty, equates it to zero and solves it, selecting the second root (solution), and assigns it to the function *chi[ ]*:

In[39]:=

```
chi[n_,x_,p_,m_,s_]=c/.
Solve[D[u[c,p]-.5 m(c-(2n+x)/3)-
Integrate[4(x-z)/(x-n)^2 m(c-z),{z,(x+n)/2,c}],c]==0,c][[1]]//
PowerExpand//Simplify
```

Out[39]=

```
                2            4. m x
(0.25 (n - 1. x)  (2. s + ----------- -
                                   2
                          (n - 1. x)

                       4. m x     2
     1. Sqrt[(2. s + -----------)  -
                                2
                     (n - 1. x)

                        2                 2
        (8. m (m (-1. n  + 2. n x + 1. x ) +

                       2                2                4
             p s (2. n  - 4. n x + 2. x ))) / (n - 1. x) ])) / m
```

For example, given a limit of 65±5, someone who would have chosen a conduct of 73 will chose a conduct of 68.29:

In[41]:=

```
chi[60,70,73,1,.1]
```

Out[41]=

```
68.2918
```

We already know from the certain violators who disregard the limit who will chose a conduct equal to the maximum of the limit:

In[34]:=

```
pmax[n_,x_,c_,m_,s_]=p/.
Solve[c==p-m/(2 s),p][[1]]//Simplify
```

Out[34]=

$$c + \frac{m}{2\ s}$$

In the case of a 65±5 limit, that will be the individuals with a preference for 75, and, *vice versa*, we verify that individuals with a preference for 75 do choose a conduct of 70:

In[40]:=

```
{pmax[60,70,70,1,.1],chi[60,70,75,1,.1]}
```

Out[40]=

```
{75., 70.}
```

We still do not know the preference at which individuals will choose the mod-point of the vagueness range (this is the limit-inducing preference). It is a simple matter to solve the conduct equation for preference and assign that the function *plow[ ]*:

In[29]:=

```
plow[n_,x_,c_,m_,s_]=p/.
Solve[clow[n,x,p,m,s]==c,p][[1]]//PowerExpand//Simplify
```

Out[29]=

$$\frac{c^2\ m + m\ n^2 + c\ (-2\ m\ n + s\ (n - x)^2)}{s\ (n - x)^2}$$

Now we can see that conduct of 65 in a limit of 65±5 will be chosen by those with a preference for 67.5, and *vice versa*:

In[32]:=

```
{plow[60,70,65,1,.1],clow[60,70,67.5,1,.1]}
```

Out[32]=

```
{67.5, 65.}
```

It is a simple matter to arrange all this information into a single function *call[ ]* which determines conduct for all preferences:

In[53]:=

```
call[n_,x_,p_,m_,s_]:=Which[
p<n,            p,
p<plow[n,x,(n+x)/2,m,s],
                clow[n,x,p,m,s],
p<pmax[n,x,x,m,s],
                chi[n,x,p,m,s],
True,           cdis[n,x,p,m,s]]
```

For example we can check the conduct of preferences for 55, 65, 73, and 76 all at once:

In[54]:=

```
{call[60,70,55,1,.1],
call[60,70,65,1,.1],
call[60,70,73,1,.1],
call[60,70,76,1,.1]}
```

Out[54]=

```
{55, 63.6603, 68.2918, 71.}
```

More importantly, we can now produce a graphical representation of the reaction to a vague limit.

In[55]:=

```
Plot[{    call[61,69,p,1,.1],
          call[64.5,65.5,p,1,.1]}
          ,{p,58,76},
PlotPoints->50,PlotDivision->50,
PlotStyle->{AbsoluteThickness[1],AbsoluteThickness[.5]},
Ticks->None];
```